

The process of deduction

Overview

What is the Prolog interpreter doing during the process of deduction (replying to a query or searching to achieve a goal)?

1. A Prolog program is consulted and the facts and rules in the program are loaded into the knowledge database.
2. When a query or goal is given, the Prolog interpreter searches the knowledge database **from top to bottom** to find any facts or rules match with the query or goal.
3. If a match is found, the search for answers to the query succeeds or the goal is true. The values of any variables found are given.
4. If the user asks for further solutions (e.g by entering a semicolon in some Prolog interpreter), the Prolog interpreter continues searching. It keeps track of the facts and the rules which have been searched. When it can find no more solutions, it says `no` and the search ends.

A detailed description of Step 2

The Prolog interpreter **starts with the goal/query and works backward by identifying the facts and the rules** (in the knowledge database) **from which the goal can be derived**. This can be done by **the expansion of a goal by applying a rule** and **the reduction of a goal by applying a fact**. You will see the process of deduction in the [example](#) below.

Unification

Unification is the derivation of a new rule from a given rule through the binding of variables.

In order to match a goal or a query, **any variable encountered is substituted with the value of an appropriate constant** (called **binding**).

Backtracking

Sometimes there are more than one fact/rule which can be applied to a goal/query. In this case, the fact/rule which appears first in the knowledge database will be applied first. The other applicable facts or rules will be applied later such that all possible solutions are found. This process is known as **backtracking**.

Examples

We will use the following program and some queries to explain the process of deduction.

Note: In each step of the process of deduction, there may be more than one subgoals to be processed. In this case, only the first subgoal is considered. If there is a rule that can be applied to it, it is replaced by those subgoals found in the applied rule. If it satisfies a fact, it is removed. If no rules or facts can be applied, we have to backtrack.

Program 4: A program about some computers, installed software and some users

```
spec(comp1, pc, 32).           /* Fact 1 */
spec(comp2, mac, 128).        /* Fact 2 */
spec(comp3, pc, 64).          /* Fact 3 */
runs(pc, movie_edit, 96).     /* Fact 4 */
runs(pc, vb, 16).             /* Fact 5 */
runs(pc, cpp, 28).            /* Fact 6 */
runs(mac, vb, 24).            /* Fact 7 */
runs(mac, prolog, 128).       /* Fact 8 */
access(judy, comp1).          /* Fact 9 */
access(peter, comp3).         /* Fact 10 */
access(david, comp1).         /* Fact 11 */
access(david, comp2).         /* Fact 12 */

can_use(P, SW) :- access(P, Comp), can_run(Comp, SW). /* Rule 1 */

can_run(Comp, SW) :- spec(Comp, CompType, MemAvail),
                      runs(CompType, SW, MemNeeded),
                      MemAvail >= MemNeeded.          /* Rule 2 */
```

Can Judy use VB?

?- can_use(judy, vb).

Level	Results of deduction	Rules or facts to be applied	Bindings of variables
Original goal	<p>can_use(judy, vb).</p> <p><i>In order that the original goal (can_use(judy, vb)) is true, according to Rule 1, the conditions in the rule must be true (i.e. access(judy, Comp) and can_run(Comp, vb) for some Comp). Therefore the goal can be reduced as shown below.</i></p> <p><i>In order to apply Rule 1 to the goal, a unification is done by binding (i.e. assigning) the value judy to the variable P and binding the value vb to the variable SW.</i></p>	<u>Rule 1</u>	P = judy SW = vb
1	<p>access(judy, Comp), can_run(Comp, vb).</p> <p><i>After unification, access(judy, comp1) is found to be true (from Fact 9). Therefore, whether the original goal is true or not depends on only whether can_run(comp1, vb) is true or not.</i></p>	<u>Fact 9</u>	Comp = comp1
2	can_run(comp1, vb).	<u>Rule 2</u>	-
3	spec(comp1, CompType, MemAvail), runs(CompType, vb, MemNeeded),	<u>Fact 1</u>	CompType = pc

	<code>MemAvail >= MemNeeded.</code>	<code>MemAvail = 32</code>
4	<code>runs(pc, vb, MemNeeded),</code> <code>32 >= MemNeeded.</code>	<u>Fact 5</u> <code>MemNeeded = 16</code>
5	<code>32 >= 16.</code> <i>The above is true, and so we can deduce that subgoals at level 4 is true, and hence those at levels 3, 2, 1 and the original goal.</i>	Satisfied! Output: yes
Overall output	yes	

Can David use Prolog?

`?- can_use(david, prolog).`

This goal demonstrates backtracking.

Level	Results of deduction	Rules or facts to be applied	Bindings of variables
Original goal	<code>can_use(david, prolog).</code>	<u>Rule 1</u>	<code>P = david</code> <code>SW = prolog</code>
1	<code>access(david, Comp), can_run(Comp, prolog).</code> <i>Both Fact 11 and Fact 12 are applicable. Since Fact 11 comes before Fact 12, Fact 11 is tried first.</i>	<u>Fact 11</u>	<code>Comp = comp1</code>
2	<code>can_run(comp1, prolog).</code>	<u>Rule 2</u>	-
3	<code>spec(comp1, CompType, MemAvail),</code> <code>runs(CompType, prolog, MemNeeded),</code> <code>MemAvail >= MemNeeded.</code>	<u>Fact 1</u>	<code>CompType = pc</code> <code>MemAvail = 32</code>
4	<code>runs(pc, prolog, MemNeeded),</code> <code>32 >= MemNeeded.</code> <i>No rules or facts can further be applied. Backtracking is done here.</i>		Failed! Backtracks.
3	<code>spec(comp1, CompType, MemAvail),</code> <code>runs(CompType, prolog, MemNeeded),</code> <code>MemAvail >= MemNeeded.</code> <i>There is no applicable rules or facts that were not applied to these subgoals before. So backtracking has to be done.</i>		Failed! Backtracks.
2	<code>can_run(comp1, prolog).</code>		Failed! Backtracks.
1	<code>access(david, Comp), can_run(Comp, prolog).</code> <i>Fact 11 has been used. Now we try Fact 12.</i>	<u>Fact 12</u>	<code>Comp = comp2</code>
2	<code>can_run(comp2, prolog).</code>	<u>Rule 2</u>	-
3	<code>spec(comp2, CompType, MemAvail),</code> <code>runs(CompType, prolog, MemNeeded),</code> <code>MemAvail >= MemNeeded.</code>	<u>Fact 2</u>	<code>CompType = mac</code> <code>MemAvail = 128</code>

4	<code>runs(mac, prolog, MemNeeded), 128 >= MemNeeded.</code>	<u>Fact 8</u> <code>MemNeeded = 128</code>
5	<code>128 >= 128.</code>	Satisfied! Output: yes
Overall output	yes	

What software can Judy use?

`?- can_use(judy, X).`

This query demonstrates how backtracking is done to find all answers.

Level	Results of deduction	Rules or facts to be applied	Bindings of variables
Original query	<code>can_use(judy, X).</code>	<u>Rule 1</u>	<code>P = judy</code>
1	<code>access(judy, Comp), can_run(Comp, X).</code>	<u>Fact 11</u>	<code>Comp = compl</code>
2	<code>can_run(compl, X).</code>	<u>Rule 2</u>	-
3	<code>spec(compl, CompType, MemAvail), runs(CompType, X, MemNeeded), MemAvail >= MemNeeded.</code>	<u>Fact 1</u>	<code>CompType = pc MemAvail = 32</code>
4	<code>runs(pc, X, MemNeeded), 32 >= MemNeeded.</code>	<u>Fact 4</u>	<code>X = movie_edit MemNeeded = 96</code>
5	<code>32 >= 96.</code>	Failed! Backtracks.	
4	<code>runs(pc, X, MemNeeded), 32 >= MemNeeded.</code>	<u>Fact 5</u>	<code>X = vb MemNeeded = 16</code>
5	<code>32 >= 16.</code> <div><i>X = vb is a solution to the original query.</i> <i>Backtracking is still performed because there may be other solutions.</i></div>	Satisfied! Output: X = vb Then backtracks.	
4	<code>runs(pc, X, MemNeeded), 32 >= MemNeeded.</code>	<u>Fact 6</u>	<code>X = cpp MemNeeded = 28</code>
5	<code>32 >= 28.</code>	Satisfied! Output: X = cpp Then backtracks.	
4	<code>runs(pc, X, MemNeeded), 32 >= MemNeeded.</code>	Failed! Backtracks.	
3	<code>spec(compl, CompType, MemAvail), runs(CompType, X, MemNeeded), MemAvail >= MemNeeded.</code>	Failed! Backtracks.	
2	<code>can_run(compl, X).</code>	Failed! Backtracks.	
1	<code>access(judy, Comp), can_run(Comp, X).</code>	Failed! Backtracks.	

Original
query

`can_use(judy, X).`

Failed!

Since the top level is reached, all answers are found.

Overall
output

`x = vb`
`x = cpp`

Who can use MovieEdit?

`?- can_use(X, movie_edit).`

Backtracking has to be performed before concluding that there is no answer to a query.

Level	Results of deduction	Rules or facts to be applied	Bindings of variables
Original query	<code>can_use(X, movie_edit).</code>	<u>Rule 1</u>	<code>SW = movie_edit</code>
1	<code>access(X, Comp), can_run(Comp, movie_edit).</code>	<u>Fact 9</u>	<code>X = judy</code> <code>Comp = comp1</code>
2	<code>can_run(comp1, movie_edit).</code>	<u>Rule 2</u>	-
3	<code>spec(comp1, CompType, MemAvail),</code> <code>runs(CompType, movie_edit, MemNeeded),</code> <code>MemAvail >= MemNeeded.</code>	<u>Fact 1</u>	<code>CompType = pc</code> <code>MemAvail = 32</code>
4	<code>runs(pc, movie_edit, MemNeeded),</code> <code>32 >= MemNeeded.</code>	<u>Fact 4</u>	<code>MemNeeded = 96</code>
5	<code>32 >= 96.</code>	Failed! Backtracks.	
4	<code>runs(pc, movie_edit, MemNeeded),</code> <code>32 >= MemNeeded.</code>	Failed! Backtracks.	
3	<code>spec(comp1, CompType, MemAvail),</code> <code>runs(CompType, movie_edit, MemNeeded),</code> <code>MemAvail >= MemNeeded.</code>	Failed! Backtracks.	
2	<code>can_run(comp1, movie_edit).</code>	Failed! Backtracks.	
1	<code>access(X, Comp), can_run(Comp, movie_edit).</code>	<u>Fact 10</u>	<code>X = peter</code> <code>Comp = comp3</code>
2	<code>can_run(comp3, movie_edit).</code>	<u>Rule 2</u>	-
3	<code>spec(comp3, CompType, MemAvail),</code> <code>runs(CompType, movie_edit, MemNeeded),</code> <code>MemAvail >= MemNeeded.</code>	<u>Fact 3</u>	<code>CompType = pc</code> <code>MemAvail = 64</code>
4	<code>runs(pc, movie_edit, MemNeeded),</code> <code>64 >= MemNeeded.</code>	<u>Fact 4</u>	<code>MemNeeded = 96</code>
5	<code>64 >= 96.</code>	Failed! Backtracks.	
4	<code>runs(pc, movie_edit, MemNeeded),</code> <code>64 >= MemNeeded.</code>	Failed! Backtracks.	

3	spec(comp3, CompType, MemAvail), runs(CompType, movie_edit, MemNeeded), MemAvail >= MemNeeded.	Failed! Backtracks.	
2	can_run(comp3, movie_edit).	Failed! Backtracks.	
1	access(X, Comp), can_run(Comp, movie_edit).	Fact 11	X = david Comp = comp1
2	can_run(comp1, movie_edit).	Rule 2	-
3	spec(comp1, CompType, MemAvail), runs(CompType, movie_edit, MemNeeded), MemAvail >= MemNeeded.	Fact 1	CompType = pc MemAvail = 32
4	runs(pc, movie_edit, MemNeeded), 32 >= MemNeeded.	Fact 4	MemNeeded = 96
5	32 >= 96.	Failed! Backtracks.	
4	runs(pc, movie_edit, MemNeeded), 32 >= MemNeeded.	Failed! Backtracks.	
3	spec(comp1, CompType, MemAvail), runs(CompType, movie_edit, MemNeeded), MemAvail >= MemNeeded.	Failed! Backtracks.	
2	can_run(comp1, movie_edit).	Failed! Backtracks.	
1	access(X, Comp), can_run(Comp, movie_edit).	Fact 12	X = david Comp = comp2
2	can_run(comp2, movie_edit).	Rule 2	-
3	spec(comp2, CompType, MemAvail), runs(CompType, movie_edit, MemNeeded), MemAvail >= MemNeeded.	Fact 2	CompType = mac MemAvail = 128
4	runs(mac, movie_edit, MemNeeded), 128 >= MemNeeded.	Failed! Backtracks.	
3	spec(comp2, CompType, MemAvail), runs(CompType, movie_edit, MemNeeded), MemAvail >= MemNeeded.	Failed! Backtracks.	
2	can_run(comp2, movie_edit).	Failed! Backtracks.	
1	access(X, Comp), can_run(Comp, movie_edit).	Failed! Backtracks.	
Original query	can_use(X, movie_edit).	Failed. The top level is reached and no answers have been found. Output: no	
Overall output	no		